
Struttura Dati
Definizione

In informatica per struttura dati si intende un modo specifico di organizzare e memorizzare i dati (elementari) in 

modo che possano essere utilizzati in modo efficiente

In altre parole una struttura dati è una definizione di dati e di operazioni che possono essere eseguite su di essi. In 

sostanza, una struttura dati fornisce un modo per organizzare e manipolare i dati in modo efficiente e significativo 

all'interno di un programma informatico.

I dati sfruttati nelle strutture dati sono quelli gestiti direttamente dall’elaboratore: byte, halfword, word, float, 

double, ascii

Le strutture dati sono progettate per ottimizzare le operazioni di accesso, modifica e ricerca dei dati

Tra le strutture dati più note ci sono: vettori, liste, code, pile, alberi, grafi. 



Struttura Dati
Classificazione

Le strutture dati possono essere statiche o dinamiche
❑ Nelle strutture dati statiche il numero dei dati è prestabilita e non muta 

(vettore o matrice)
❑ Nelle strutture dati dinamiche il numero di dati cambia nel corso 

dell’esecuzione del programma (pila, coda, lista, grafo). In questo ultimo 
caso un campo dei dati può essere usato  per contenere l’indirizzo di 
un’ulteriore elemento della struttura dati (si dice che punta o che è un 
puntatore a quest’ultima)



VETTORE



VETTORE
Generalità

Un vettore (o array) è un insieme ordinato di elementi tutti dello stesso tipo disposti 

sequenzialmente in memoria centrale

In informatica i vettori possono essere utilizzati in diversi contesti e offrono molteplici vantaggi:

❑ Memorizzazione sequenziale: i vettori memorizzano elementi in modo sequenziale in memoria, 

consentendo un accesso rapido agli elementi tramite un indice

❑ Accesso casuale: Gli elementi di un vettore possono essere accessibili in modo casuale tramite 

l'indice corrispondente, il che rende l'accesso ai dati efficiente

❑ Implementazione di strutture dati complesse: i vettori sono spesso utilizzati come base per 

implementare altre strutture dati complesse come liste, code e pile.

❑ Algoritmi di elaborazione dati: Molte operazioni di elaborazione dati, come ordinamento, 

ricerca e manipolazione, possono essere implementate efficientemente utilizzando vettori



VETTORE
Definizione fisica

Un vettore è definito 
esplicitando il numero di 
elementi che lo compongono 
(lunghezza del vettore)

L’accesso ad un elemento del 
vettore è consentito 
specificando un indice

Il primo elemento del 
vettore ha indice 0

Indice Indirizzo in 
memoria

Valore in 
memoria

0 100 12

1 104 34

2 108 -121

3 112 0

4 116 54



VETTORE
Definizione in un linguaggio ad alto livello

Un vettore è definito, in un 
linguaggio ad alto livello, dalla 
specifica del tipo, dall’identificatore 
e dalla sua lunghezza (cioè del 
numero di elementi)

È possibile inizializzare un elemento 
di un vettore specificando 
l’identificativo e l’indice (nel 
linguaggio C si usano le parentesi 
quadre) e il valore

Per prelevare un elemento è 
sufficiente utilizzare l’identificativo
e l’indice (e una variabile per la 
memorizzazione del dato)

Dichiarazione int array[5]; Crea un vettore di 
interi in memoria 
(cinque elementi 
consecutivi)

Inizializzazione array[0]=45; Inserisce nella prima 
locazione di memoria 
assegnata all’array il 
valore 45

Recupero dati int temp=array[0] Nella variabile temp si 
trova il valore relativo 
alla prima locazione di 
memoria associata 
all’array (ovvero 
temp=45)



VETTORE
Manipolazione usando un linguaggio ad alto livello

Effettuare la somma degli elementi 
di due vettori (di lunghezza 3) a 
medesima posizione

ESEMPIO

INPUT:
V1=(13,24,35)
V2=(47,16,-25)

OUTPUT:
V=(60,40,10)

#include <stdio.h>

int main()

{

const int dim_vett=3;

int i, a,b,addition;

int v1[dim_vett]={13,24,35};

int v2[dim_vett]={47,16,-25};

int v3[dim_vett];

for (i=0;i<dim_vett;i=i+1){

a=v1[i];

b=v2[i];

addition=a+b;

v3[i]=addition;

}

for (i=0;i<dim_vett;i=i+1){

printf("%d\t", v3[i]);

}

return 0;

}



VETTORE IN MARS
Generalità

❑Un vettore è un sequenza di dati 
omogenei (dello stesso tipo) 
disposti sequenzialmente in 
memoria

❑I dati o elementi di un array 
hanno tutti la stessa 
dimensione di tipo (esize o 
dimension) correlata al tipo 
elementare: word sono 4byte, 
half sono 2byte, byte sono 
8bit per elemento

❑La posizione (indice) 
identifica i singoli elementi

Indice Indirizzo in 
memoria

Valore in 
memoria

0 100 12

1 104 34

2 108 -121

3 112 0

4 116 54



VETTORE IN MARS
Definizione

❑In MARS un vettore si definisce 
specificando il tipo e interponendo 
delle virgole tra gli elementi

❑Essendo una struttura statica in  
MARS è doveroso riportare anche il 
numero di elementi che lo 
costituiscono (lunghezza del 
vettore)

Esempio:
array8bit: .byte 12,34,-121,0,54

lunghezzaarray8bit: .byte 5

Indice Posizione Valore in memoria

0 100 12

1 101 34

2 102 -121

3 103 0

4 104 54



VETTORE IN MARS
Definizione

Esempio:
array16bit: .half 12000,-2000,-15,78

lunghezzaarray16bit: .byte 4

Indice Posizione Valore in 
memoria

0 200 12000

1 202 -2000

2 204 -15

3 206 78

Indice Posizione Valore in 
memoria

0 300 67067

1 304 23400000

2 308 11

3 312 -785

Esempio:
array32bit: .word 67067,23400000,11,-785

lunghezzaarray32bit: .byte 4



VETTORE IN MARS
Inizializzazione

❑È possibile inizializzare un vettore ad un 
valore specifico definendo il valore e la sua 
lunghezza (numero di elementi) con una 
sintassi del tipo

etichetta: .tipo valore_di_inizializzazione:numero_elementi

❑Esempio:
vettore: .word 2:10

Posizione Valore in memoria

0 2

4 2

8 2

12 2

16 2

20 2

24 2

28 2

32 2

36 2



VETTORE IN MARS
Individuazione di un elemento

❑ L’indirizzo dell’elemento è ottenuto da:

base + index* dimension

dove 
base è l’indirizzo di memoria dove inizia il vettore

index è l’incide dell’elemento che si vuole 
manipolare (inizializzare o prelevare)

dimension è la dimensione degli elementi (dipende 
dal tipo elementare)

Posizione Valore in memoria

0 2

4 2

8 2

12 2

16 2

20 2

24 2

28 2

32 2

36 2



VETTORE IN MARS
Individuazione di un elemento

❑ L’indirizzo del quarto elemento è ottenuto da:

base + index* dimension

dove 
base è 100

index è 3

dimension è 4

Cioè locazione di memoria: 112

Indice Valore in memoria

0 Array >100 123

1 104 56

2 108 44

3 112 22

4 116 -33

5 120 67

Esempio:
array32bit: .word 123,56,44,22,-33,67

lunghezzaarray32bit: .byte 6



VETTORE IN MARS
Manipolazione dei dati

Per accedere agli elementi di un vettore in lettura e in scrittura si può ricorrere a 
due modalità

ACCESSO

A registro con 
spiazzamento 

simbolico

Spiazzamento a 
registro



VETTORE IN MARS
Manipolazione dei dati:  a registro con spiazzamento simbolico

❑Per accedere agli elementi di un 
vettore in lettura e in scrittura si può 
ricorrere al modo di indirizzamento 
a registro con spiazzamento 
simbolico

vettore($a0)

Dove $a0 deve contenere la 
posizione dell’elemento moltiplicato 
per la sua dimensione  (4 in caso di 
word, 2 per le halfword, 1 per il 
byte)

li $t1,0 #indice relativo al primo elemento

mul $a0,$t1,4 #indice*dimensione (posizione relativa)

lw $t0,array($a0) # copia/prelievo del primo elemento

#il calcolo su array($a0) riporta

#la posizione assoluta

li $t1,3 #indice relativo al quarto elemento

mul $a0,$t1,4 #posizione relativa del quarto elemento

lw $t3,array($a0) #prelievo del quarto elemento

.data

array: .word 45,100,-23,56

lunghezzaarray: .byte 4



VETTORE IN MARS
Manipolazione dei dati

Stampa di un vettore lungo 5 elementi di dimensione 16bit



VETTORE IN MARS
Manipolazione dei dati

.text

.globl main

main:

# STAMPA DI UN VETTORE

lw $t1,nunelem # $t1 numero elementi del vettore

li $t2,0 # $t2 indice

loop: 

mul $t3,$t2,2 # indice*dimensione (moltiplicare x 2 perché halfword)

lh $t4,array($t3)   # copia dell’i-esimo elemento nel registro $t4 cioè $t4=v[i]

move $a0,$t4      #sposto l’elemento per effettuarne la stampa

li $v0,1 #stampo i-esimo elemento

syscall

la $a0,tabulato  # Stampo un tabulato (per una maggiore leggibilità)

li $v0,4

syscall

add $t2,$t2,1  # incremento indice

blt $t2,$t1,loop #confronto per determinare la fine del vettore

li $v0,10

syscall

.data

array: .half 12,43,23,54,77

nunelem: .word 5

tabulato: .asciiz "\t"



VETTORE IN MARS
Manipolazione dei dati:  a registro con spiazzamento

❑Un’altra possibilità è quella di 
utilizzare come modo di 
indirizzamento per accedere agli 
elementi a registro con spiazamento

($a0)

dove $a0 deve contenere un indirizzo 
costituito da

posizione_vettore+ (posizione_elemento)*dimensione

li $t1,0 # indice relativo del primo elemento

mul $t1,$t1,4 # indice*dimensione (posizione relativa)

la $a0,array   # indirizzo di inizio del vettore

add $a0,$a0,$t1

lw $t0,($a0)   # copia del primo elemento

li $t1,3     # indice relativo del IV elemento

mul $t1,$t1,4 # indice*dimensione (posizione relativa)

add $a0,$a0,$t1 # aggiornamento  posizione fisica del IV 

#elemento

lw $t1, ($a0)  # copia/prelievo del quarto elemento

.data

array: .word 45,100,-23,56

lunghezzaarray: .byte 4



VETTORE IN MARS
Manipolazione dei dati

Stampa di un vettore lungo 5 elementi di dimensione 16bit



VETTORE IN MARS
Manipolazione dei dati

.text

.global main

main:

la $t0,array # locazione di inizio del vettore

lw $t1,nunelem # $t1 numero elementi del vettore

li $t2,0 # $t2 indice

loop: 

mul $t3,$t2,4 # indice * dimensione

add $t3,$t0,$t3 #posizione+(indice*dimensione)

lh $t4,($t3)   # copia dell’i-esimo elemento in $t4

move $a0,$t4     #... Elaboro elemento (es.:stampa)

li $v0,1

syscall

la $a0,tabulato  # Stampo un tabulato per una maggiore leggibilità

li $v0,4

syscall

addu $t2,$t2,1  # incremento indice

blt $t2,$t1,loop

li $v0,10

syscall

.data

array: .half 12,43,23,54,77

nunelem: .word 5

tabulato: .asciiz "\t"

Si 
guadagna 

in 
efficienza



ESERCIZIO

Definire due vettori di 5 elementi x di valore -216<xi<215-1 e eseguire il prodotto 
scalare (⊥)

NB: 
v1=(3,5,8,10,1) e v2=(1,2,3,0,13) 

v1⊥v2= (31)+(52)+(83)+(100)+(113)=50



ESERCIZIO

Prodotto scalare

i=0
prodscal=0

numelem=5

a=v1[i]
b=v2[i]

prodscal=prodscal + (a*b)

i++

i<numelem

STAMPA prodscal



ESERCIZIO
.text

.globl main

main:

xor $t0,$t0,$t0  #inizializzazione a zero del registro contenete l’indice del vettore

li $t5,0  #inizializzatore del registro che contiene il prodotto scalare

lw $t9,numelem #lunghezza del vettore

ciclo:

mul $t1,$t0,2 #spiazzamento all’i-esimo elemento indice*dimensione

lh $t2,v1($t1) #v1[i]

lh $t3,v2($t1) #v2[i]

mul $t4,$t2,$t3 #v1[i]*v2[i]

add $t5,$t5,$t4 #prodotto scalare parziale

add $t0,$t0,1 #incremento indice

bne $t0,$t9,ciclo #confronto fine vettore

move $a0,$t5 #stampa del prodotto scalare

li $v0,1

syscall

.data

v1: .half 3,5,8,10,1

v2: .half 1,2,3,0,13

numelem:.word 5



ESERCIZIO II

❑Definire un vettore v1 di 6 elementi x di valore -232<xi<231-1 e memorizzarlo in un 
nuovo vettore includendo solo gli elementi multipli di 5. Stampare v2

INPUT
v1=(3,5,8,10,4,20) 

OUTPUT
v2= (5,10,20)



ESERCIZIO II
i=0
j=0

a=v1[i]

a%5==0

i=i+1

i<numelem

STAMPA v2

v2[j]=v1[i]

j=j+1



ESERCIZIO II

.data

v1: .word 3,5,8,10,1,7

v2:.space 24

numelem:.word 6

separatore:.asciiz ″\t″

stampa:

lw $a0,($t1)

li $v0,1

syscall

la $a0,separatore

li $v0,4

syscall

addu $t1,$t1,4

blt $t1,$t8, stampa

li $v0,10

syscall

.text

.globl main

main:

la $t0,v1 # indirizzo al primo elemento di v1

la $t1,v2 # indirizzo al primo elemento di v2

lw $t2,numelem    # $t2 numero elementi del vettore 

mul $t9,$t2,4       # spiazzamento dell’ultimo elemento di v1

add $t9,$t9,$t0    # indirizzamento dell’ultimo elemento di v1

loop: 

lw $t3,($t0)    #copia dell’i-esimo elemento di v1 nel registro $t3

rem $t4,$t3,5 # vedo se v1[i] è multiplo di 5

bnez $t4,salta

sw $t3,($t1) #memorizzo v2[j]=v1[i]

addu $t1,$t1,4 # incremento indice v2

salta:

addu $t0,$t0,4 # incremento indice v1

bgt $t9,$t0,loop #ciclo se non ho letto tutto  v1

move $t8,$t1 # dimensione v2

la $t1,v2 # lettura posizione inziale v2

Si calcola 
l’indirizzo relativo all’inizio del vettore
per poi svolgere il salto



STRINGA



STRINGA
Generalità

Una stringa è una sequenza di caratteri, come lettere, numeri o simboli, che è 
trattata come un singolo dato

Le stringhe sono comunemente utilizzate per rappresentare testo e sono 
ampiamente utilizzate nei linguaggi di programmazione per manipolare e 
memorizzare dati di testo

Le operazioni comuni sulle stringhe includono la ricerca, l'estrazione di 
sottostringhe, la concatenazione (unione) di due o più stringhe e la sostituzione 
di caratteri



STRINGA
Generalità

In campo informatico una stringa è una sequenza di numero che rappresentano 
dei simboli, caratteri e cifre più simboli di controllo

Ad esempio le lettere dell’alfabeto possono essere rappresentate dai numeri 
A=1, B=2,C=3,…

Nel 1963 l'American National Standards Institute (ANSI) decretò la nascita 
dell'American Standard Code for Information Interchange (ASCII) per codificare 
le lettere dell'alfabeto statunitense, le cifre e simboli di ausilio.



STRINGA
Codifica utilizzata – STANDARD ASCII



STRINGA
Definizione in un linguaggio ad alto livello

Una stringa è definita, in un 
linguaggio ad alto livello, dal tipo 
char, dall’identificatore e della sua 
lunghezza (cioè del numero di 
caratteri)

È possibile inizializzare un elemento 
di una stringa specificando 
l’identificativo e l’indice (nel 
linguaggio C si usano le parentesi 
quadre) e il simbolo

Per prelevare un elemento è 
sufficiente utilizzare l’identificativo
e l’indice (e una variabile per la 
memorizzazione del dato)

Dichiarazione char stringa[5]; Crea una stringa (cinque 
simboli consecutivi in 
memoria)

Inizializzazione stringa="Ciao"; Inizializza con la stringa 
con le lettere Ciao

Recupero dati char simbolo=stringa[2] Nella variabile temp si 
trova il carattere ‘a’



STRINGA
Manipolazione usando un linguaggio ad alto livello

Effettuare il calcolo del numero di 
volte in cui si ripete la lettera A in 
una stringa lunga 9 caratteri

ESEMPIO

INPUT:
"Alabarda"

OUTPUT:
3

#include <stdio.h>

int main()

{

char stringa[9]="Alabarda";

int count,i;

count=0;

for(i=0;i<9;i++){

if (stringa[i]=='a'){count=count+1;}

}

printf("Il numero di a presenti nella stringa %s è: %d",stringa,count);

return 0;
}



STRINGA IN MARS
Generalità

❑Una stringa è un vettore di valori 
numerici, di un byte, che hanno il 
significato di carattere 
alfanumerico

❑Le stringhe sono sequenze 
numeriche (array) di caratteri con un 
terminatore NULL (il valore 0)
❑La stringa priva di caratteri , la stringa 
vuota, è quella formata dal solo 
terminatore NULL (tipicamente valore 
0)

❑I valori in codice ASCII prevedono 
le lettere maiuscole [A,Z] con 
intervallo[65,90]; minuscole [a,z] 
con intervallo [97,122]; e le cifre 
[0,9] con intervallo [48,57]

Indice Indirizzo in 
memoria

Valore in 
memoria

Rappresentazione

0 100 67 C

1 101 65 A

2 102 83 S

3 103 65 A

4 104 0 0



STRINGA IN MARS
Definizione

❑In MARS una stringa si definisce 
specificando il tipo asciiz o ascii e 
interponendo dei doppi apici tra la 
sequenza di caratteri

❑Pur essendo una struttura statica in  
MARS NON è doveroso riportare il numero 
di elementi che lo costituiscono 
(lunghezza del vettore) perché se si usa la 
direttiva asciiz l’assemblatore pone il 
valore 0 alla fine dei caratteri tra doppi 
apici (e quindi è nota la terminazione della 
stringa; il valore zero è detto anche NULL o 
terminatore di stringa)

Esempio:
stringa1: .asciiz ″Cane″

stringa2: .ascii ″Gatto″



STRINGA IN MARS
Definizione

Esempio:
stringa1: .asciiz ″Cane″

Indice Posizione Valore in 
memoria

Rappresentazione

0 200 67 C

1 201 97 a

2 202 110 n

3 203 101 e

4 204 0

Esempio:
stringa2: .ascii ″Gatto″

Indice Posizione Valore in 
memoria

Rappresentazione

0 200 71 G

1 201 97 a

2 202 116 t

3 203 116 t

4 204 111 o

5 205 ? ?



STRINGA IN MARS
Individuazione di un elemento

❑ L’indirizzo di un carattere è ottenuto da:

base + index* dimension

dove 
base è l’indirizzo di memoria dove inizia il vettore

index è l’incide dell’elemento che si vuole 
manipolare (inizializzare o prelevare)

dimension è la dimensione degli elementi cioè 1 
perché si tratta di byte

Posizione Valore in memoria

0 G

1 a

2 t

3 t

4 i

5 n

6 a

7 0



STRINGA IN MARS
Manipolazione dei dati

Per accedere agli elementi di una stringa in lettura e in scrittura si può ricorrere a 
due modalità

ACCESSO

A registro con 
spiazzamento 

simbolico

Spiazzamento a 
registro

lb $t1, stringa($t0) la $t0,stringa
lb $t1,($t0)



STRINGA
ESEMPIO I

Calcolare il numero di caratteri di una stringa (str_len function)

ESEMPIO

INPUT:
"Dedicato a chi crede che la colonna sonora quando ci sbatti contro… suoni"

OUTPUT:75



.text

.globl main

main:

xor $t1,$t1,$t1 #contatore delle lettere

xor $t0,$t0,$t0 #indice stringa

ciclo:

lb $t9,stringa($t0) #Lettura i-esimo carattere

beqz $t9, esci #confronto con terminatore nullo

add $t0,$t0,1 #incremento indice

add $t1,$t1,1 #incremento contatore

j ciclo

esci:

move $a0,$t0 #stampa la lunghezza della stringa

li $v0,1 #

syscall #

li $v0,10

syscall

STRINGA
ESEMPIO I

.data

stringa: .asciiz "Dedicato a chi crede che la colonna sonora quando ci sbatti contro… suoni"



STRINGA
ESEMPIO II

Effettuare il calcolo del numero di volte in cui si ripete la lettera a in una stringa

ESEMPIO

INPUT:
"Dedicato a chi non ha mai letto Non Do Maho, la storia di un ricco avaro 
giapponese"

OUTPUT:
10



.text

.globl main

main:

xor $t2,$t2,$t2 #contatore delle lettere 'a'

li $t8,97 #valore ASCII della lettera 'a' 

#in alternativa 

#lb $t8,lettera_a

la $t0,stringa #Indirizzo di inizio della stringa in $t0

ciclo:

lb $t9,($t0) #Lettura i-esimo carattere

bne $t9 $t8, non_trovata #confronto

add $t2,$t2,1 #incremento contatore

non_trovata:

add $t0,$t0,1 #incremento indirizzo assoluto

bnez $t9, ciclo #si continua il ciclo se non si raggiunge la fine della stringa cioè il valore 0

move $a0,$t2 #stampa del numero di 'a'

li $v0,1 #

syscall #

li $v0,10

syscall

STRINGA
ESEMPIO II

.data
stringa: .asciiz "Dedicato a chi crede che la colonna sonora quando ci sbatti contro… suoni"
lettera_a:.ascii "a"



STRINGA
ESEMPIO III

Realizzare la funzione UPPERSTRING che trasforma ogni carattere di una stringa 
minuscolo in maiuscolo

ESEMPIO

INPUT:
"Dedicato a chi si rifa’ la bocca con magia LabbraCadabbra"

OUTPUT:
"DEDICATO A CHI SI RIFA’ LA BOCCA CON MAGIA… LABBRACADABBRA" 



.text

.globl main

main:

la $t0,stringa #Indirizzo di inizio della stringa in $t0

ciclo:

lb $t9,($t0) #Lettura i-esimo carattere

bge $t9,97, minuscolo1 #prima analisi se è un carattere minuscolo

j non_minuscolo

minuscolo1:

ble $t9,122, minuscolo2 #seconda analisi se è un carattere minuscolo

j non_minuscolo

minuscolo2:

sub $t9,$t9,32 #associazione con la lettera maiuscola

sb $t9,($t0) #conservazione del carattere minuscolo modificato

non_minuscolo:

beqz $t9,fine #analisi della terminazione della stringa

add $t0,$t0,1 #incremento dell'indice della stringa

j ciclo

fine:

la $a0,stringa #stampa della stringa risultante

li $v0,4 #

syscall #

li $v0,10

syscall

STRINGA
ESEMPIO III

.data
stringa: .asciiz "Dedicato a chi si rifa’ la bocca con magia LabbraCadabbra "



STRINGA
ESEMPIO IV

Leggere una stringa da tastiera e stampare la stringa immessa e la sua rappresentazione 
inversa

ESEMPIO
INPUT:
"Dedicato a chi sa che il prezzo del mondo è tutto compreso eccetto gli extra che sono 
terrestri"
OUTPUT:
"irtserret onos ehc artxe ilg ottecce oserpmoc ottut è odnom led ozzerp li ehc as ihc a 
otacideD" 



.text

.globl main

main:

la $a0,stringa #Lettura stringa da tastiera (lunghezza impostata a 255 per default

li $a1,255 #IMPORTANTE: NEL MARS SI ACCODA IL RITORNO A CAPO \n DOPO 

li $v0,8 #L’INSERIMENTO DELLA STRINGA E LA PRESSIONE DEL TASTO  OK

syscall

#Calcolo della lunghezza della stringa

la $t0,stringa #lettura indirizzo della stringa letta

ciclo: #

lb $t1,($t0) #

beqz $t1,fine_calcolo_lung #

add $t0,$t0,1 #

j ciclo #

fine_calcolo_lung:

sub $t0,$t0,1 #eliminazione del simbolo di andata a capo

move $t9,$t0 #alla fine del ciclo in $t9  ci sarà la posizione assoluta dell'ultimo elemento

la $t0,stringa #posizione primo elemento della lista

scambia:

lb $t2,($t0) #lettura i elemento (primo valore di i=0)

lb $t3,($t9) #lettura j elemento (primo valore di j=n)

sb $t3,($t0) #scambio lettere $t2.->pos[j]  e $t3->pos[i]

sb $t2,($t9) #

add $t0,$t0,1 # Incremento i

sub $t9,$t9,1 # Decremento j

blt $t0,$t9, scambia

la $a0,stringa #Stampa stringa inversa

li $v0,4

syscall

STRINGA
ESEMPIO IV

.data

stringa: .space 255



MATRICE



MATRICE
(Generalità)

Una matrice è una tabella di dati omogenei (dello stesso tipo) costituita da 
r righe ed c colonne Mrxc

❑I dati o elementi di una matrice hanno tutti la stessa dimensione (esize)
❑La posizione (indice riga/ indice colonna) identifica i singoli elementi

a11 a12 a13 a14

a21 a22 a23 a24

a31 a32 a33 a34

a41 a42 a43 a44

a51 a52 a53 a54

1 32 81 12

123 490 72 345

100 58 63 44

9 34 556 33

8 56 77 22



MATRICE
La matrice nel MARS

Il MIPS non consente una particolare definizione 
di matrice: si utilizza un vettore con lunghezza 
mxn e un indice per accedere alla i-esima 
posizione

a11 a12 a13 a14

a21 a22 a23 a24

a31 a32 a33 a34

a41 a42 a43 a44

a51 a52 a53 a54

a11

a12

a13

a14

a21

a22

a23

a24

a31

a32

a33

a34

a41

a42

a43

a44

a51

a52

a53

a54



MATRICE
Definizione

Il MIPS non consente una particolare definizione di matrice. Per questo nella sua 
definizione va specificato il numero di righe il numero di colonne e gli elementi della 
matrice (quest’ultima alla stessa stregua di un vettore). Si usano due indici per scorrere gli 
elementi della matrice

.data

nrig: .word 10  # definizione del valore del numero di righe

ncol: .word 10 # definizione del valore del numero di colonne

matrice: .word 0:100 # area dell’area di memoria che conterrà la matrice 

#100 words inizializzate a zero

.text

li $t0,0 # in $t0 viene posto l'indice di riga i che assume valori da 0 a 9 

li $t1,0 # in $t1 viene posto l'indice di colonna j che assume valori da 0 a 9 

lw $t2,nrig # in $t0 viene posto il numero di righe della matrice

lw $t3,ncol # in $t1 viene posto il numero di colonne della matrice



MATRICE
Indici

Gli indici relativi della matrice iniziano dal valore 0 che corrisponde all’elemento (0,0)

Gli indici relativi devono essere moltiplicati per la dimensione dell’elemento 
determinato dal tipo (4 per word, 2 per half, 1 per byte) e sommati alla posizione 
della prima locazione in cui inizia il vettore che identifica la matrice

a00 a01 a02 a03

a10 a11 a12 a13

a20 a21 a22 a23

a30 a31 a32 a33

a40 a41 a42 a43

0 1 2 3

4 5 6 7

8 9 10 11

12 13 14 15

16 17 18 19



MATRICE
Indici

Esempio: Matrice halfword

a00 a01 a02 a03

a10 a11 a12 a13

a20 a21 a22 a23

a30 a31 a32 a33

a40 a41 a42 a43

0 2 4 6

8 10 12 14

16 18 20 22

24 26 28 30

32 34 36 38



MATRICE
Indici

Esempio: Matrice word

a00 a01 a02 a03

a10 a11 a12 a13

a20 a21 a22 a23

a30 a31 a32 a33

a40 a41 a42 a43

0 4 8 12

16 20 24 28

32 36 40 44

48 52 56 60

64 68 72 80



MATRICE
Inizializzazione in MARS

È possibile inizializzare una matrice ad un valore specifico definendo il 
valore e la sua dimensione con una sintassi del tipo

etichetta: .tipo valore_di_inizializzazione:numero_elementi

Esempio:

matrice: .word 2:20

(matrice 5x4))

2 2 2 2

2 2 2 2

2 2 2 2

2 2 2 2

2 2 2 2



MATRICE
Inizializzazione in MARS

In memoria la matrice inizializzata avrà 

l’organizzazione fisica in forma vettoriale

etichetta: .tipo valore_di_inizializzazione:numero_elementi

Esempio:

matrice: .word 2:20

(matrice 5x4))

a11 2

a12 2

a13 2

a14 2

a21 2

a22 2

a23 2

a24 2

a31 2

a32 2

a33 2

a34 2

a41 2

a42 2

a43 2

a44 2

a51 2

a52 2

a53 2

a54 2



MATRICE
Individuazione di un elemento della matrice

Data una matrice ad R righe e C colonne, come si fa ad individuare a 
quale elemento corrisponde (r,c) se si usa la notazione classica con prima 
cella a(1,1)?

a11 a12 a13 a14

a21 a22 a23 a24

a31 a32 a33 a34

a41 a42 a43 a44

a51 a52 a53 a54

Posizione elemento 
ar,c= C(r-1)+(c-1)

Es: a43=4(3)+2=14



MATRICE
Esempio I

for (i=1;i<R;i++)
{
for (j=1;j<C;j++)

{
elemento= M[i,j]
print (elemento);
print(“\t”)
}

print (“\n”);
}

Stampa elementi di una matrice



MATRICE
Esempio I

.text

.globl main

main:

li $t0,1 #indice r

li $t1,1 #indice c

lw $t2,R #numero righe R

lw $t3,C #numero colonne C

analisi_riga:

li $t1,1

analisi_colonna:

sub $t6,$t0,1 #calcolo elemento r,c

mul $t9,$t6,$t3 #

sub $t7,$t1,1 #r,c= C(r-1)+(c-1)

add $t9,$t9,$t7 #

lb $t8,matrice($t9) #prelievo elemento

move $a0,$t8 #stampa elemento

li $v0,1 #

syscall #

la $a0,tabulato #stampa tabulato

li $v0,4

syscall

addi $t1,$t1,1 #incremento colonna

ble $t1,$t3, analisi_colonna

la $a0,riga #stampa andata a capo (nuova riga)

li $v0,4

syscall

addi $t0,$t0,1 #incremento riga

ble $t0,$t2, analisi_riga
li $v0,10

syscall

.data

matrice: .byte 2:20

R: .word 4

C: .word 5

tabulato:.asciiz "\t"

riga:.asciiz "\n"



MATRICE
Esercizio Proposto

Data una matrice 4x4 di interi (word) riportare la somma degli elementi presenti 
lungo le diagonali

a11 a12 a13 a14

a21 a22 a23 a24

a31 a32 a33 a34

a41 a42 a43 a44



MATRICE
Esercizio Proposto

Data una matrice 4x4 di interi (word) riportare la somma degli elementi presenti 
lungo le diagonali

45 2534643 34453 3

56474 6 337 8232

943634 670 104 23415

60000 67324 432435 10000

10155+61010=71165



MATRICE
Esercizio Propostoxor $s0,$s0,$s0 #contatore

li $t0,1 #indice r

li $t1,1 #indice c

lw $t2,R #numero righe R

lw $t3,C #numero colonne C

ciclo1: #analisi diagonale principale (sin->dex)

sub $t6,$t0,1 #calcolo elemento r,c

mul $t9,$t6,$t3 #

sub $t7,$t1,1 # r,c= C(r-1)+(c-1)

add $t9,$t9,$t7 #

mul $t9,$t9,4 # dimensione

lw $t8,matrice($t9) #prelievo elemento

add $s0,$s0,$t8 #somma al contatore

add $t0,$t0,1 #incremento riga

add $t1,$t1,1 #incremento colonna

ble $t1,$t3,ciclo1 #confronto con fine colonna (i,j)=(R,C)

li $t0,1 #indice r

li $t1,4 #indice c

lw $t2,R #numero righe R

lw $t3,C #numero colonne C

ciclo2: #analisi diagonale secondaria

sub $t6,$t0,1 #calcolo elemento r,c

mul $t9,$t6,$t3 #

sub $t7,$t1,1 # r,c= C(r-1)+(c-1)

add $t9,$t9,$t7 #

mul $t9,$t9,4 # dimensione

lw $t8,matrice($t9) #prelievo elemento

add $s0,$s0,$t8 #somma al contatore

add $t0,$t0,1 #incremento riga

sub $t1,$t1,1 #decremento colonna

ble $t0,$t2,ciclo2 #confornto con fine (i,j)=(R,1)

move $a0,$s0

li $v0,1

syscall
.data
matrice: .word 45,2534643,34453,3,56474,6,337,8232,943634,670,104,23415,60000,67324,432435,10000
R: .word 4
C: .word 4

